

## Paper SD-11

# GEN\_OMEGA2: A SAS<sup>®</sup> Macro for Computing the Generalized Omega-Squared Effect Size Associated with Analysis of Variance Models

Anh P. Kellermann, Jeanine L. Romano, Patricia Rodríguez de Gil, Thanh Pham, Patrice Rasmussen, Yi-Hsin Chen, Yong Li, Jeffrey D. Kromrey  
University of South Florida

## ABSTRACT

Effect sizes are strongly encouraged to be reported in addition to statistical significance and should be considered in evaluating the results of a study. The choice of an effect size for ANOVA models can be confusing because indices may differ depending on the research design as well as the magnitude of the effect. Olejnik and Algina (2003) proposed the generalized eta-squared and omega-squared effect sizes which are comparable across a wide variety of research designs. This paper provides a SAS macro for computing the generalized omega-squared effect size associated with analysis of variance models by utilizing data from PROC GLM ODS tables. The paper provides the macro programming language, as well as results from an executed example of the macro.

**Keywords:** Effect sizes, ANOVA, Generalized Eta-Squared, Generalized Omega-Squared, BASE SAS, PROC GLM ODS TABLES

## INTRODUCTION

Long gone are the days when social and behavioral science researchers should simply report obtained test statistics (i.e.,  $t$ ,  $F$ ,  $\chi^2$ ) and their corresponding  $p$ -values. Over the years, interpreting the importance of scientific research based on the dichotomous reject or fail-to-reject decision has become less popular among some disciplines such as psychology and education. Instead, researchers are encouraged to supplement hypothesis test results with measures of effect magnitude. In fact, according to Thompson (2005), 24 peer-reviewed journals had explicit editorial policies that required authors to include effect sizes or other measures of effect magnitude. A large part of this “cultural change” began over a decade ago when the APA Task Force on Statistical Inference issued their statement that researchers should regularly report effect sizes, calculate confidence intervals, and use graphics to better communicate the nature of their findings for all primary outcomes (Wilkinson & APA Task Force on Statistical Inference, 1999). Unlike  $p$ -values that are used to determine if an observed effect or relationship is real or due to chance or sampling variability, effect sizes are used to estimate how large the effect or relationship is. Thus, when used together, not only can researchers make statements about the statistical significance of their findings, but, they can also report on the practical significance of their findings.

## EFFECT SIZES COMMONLY USED WITH ANALYSIS OF VARIANCE MODELS

Commonly reported effect sizes for ANOVA models include  $\eta^2$ , partial  $\eta^2$ ,  $\omega^2$ , and partial  $\omega^2$ . In general, all four of these effect sizes represent measures of association. However, there are important differences among them. First, both  $\eta^2$  and partial  $\eta^2$  are sample effect size estimates, representing the proportion of sample variability in the dependent variable that is associated with variability in an independent variable. These statistics, however, are positively biased as point estimates of the population effect size. The effect sizes  $\omega^2$  and partial  $\omega^2$  are adjusted to provide better population estimates.

Second, partial and non-partial estimates represent different measures of association. For example, consider a two-way, balanced factorial ANOVA with independent variables of student gender and grade level, and a dependent variable of mathematics achievement. The analysis of variance will provide sums-of-squares (and mean squares) for each of the two main effects (main effects for gender and grade level), the interaction effect, and the within-cell residual or error. The total sums-of-square, representing all of the sample variability in mathematics achievement, is simply the sum of these four components. The formula for  $\eta^2$  associated with the gender main effect is

$$\eta^2_{gender} = \frac{SS_{gender}}{SS_{total}}$$

In contrast, the formula for partial  $\eta^2$  for this effect is

$$partial \eta^2_{gender} = \frac{SS_{gender}}{SS_{gender} + SS_{error}}$$

As is evident in the first formula,  $\eta^2_{gender}$  represents the proportion of the total sample variance in math achievement that is associated with gender. In contrast, partial  $\eta^2_{gender}$  represents the proportion of variance in math achievement associated with gender, after the effects of grade level and the interaction between gender and grade level have been removed (note the different denominators for  $\eta^2_{gender}$  and partial  $\eta^2_{gender}$ ). Analogous formulas are used to calculate sample effect sizes for the other two sources of systematic variance in this design – grade level and the interaction between gender and grade level.

Next, to obtain a relatively unbiased estimate of the variance explained in the population by an independent variable, omega-squared can be calculated. Using the same example as above, the formula for  $\omega^2$  is presented below.

$$\omega^2_{gender} = \frac{SS_{gender} - (k-1)MS_{error}}{SS_{total} + MS_{error}}$$

Where

$k - 1$  = the degrees of freedom for the independent variable.

In this formula,  $\omega^2_{gender}$  represents the proportion of the population variance in math achievement associated with gender. However, unlike eta-squared, omega-squared takes random error ( $MS_{error}$ ) into account. Through this process, omega-squared values will be smaller than eta-squared values, with more noticeable differences occurring with smaller samples and/or research designs that include more independent variables.

Similarly, partial omega-squared represents an unbiased estimate of the population proportion of variance in math achievement associated with gender, after the effects of grade level and the interaction between gender and grade level have been removed (note the different denominators for  $\omega^2_{gender}$  and partial  $\omega^2_{gender}$ ).

$$partial\ \omega^2_{gender} = \frac{SS_{gender} - (k-1)MS_{error}}{SS_{gender} + (N - k - 1)MS_{error}}$$

Unfortunately, all four of these effect sizes can be influenced by the research design that underlies the ANOVA model (see, for example, Cohen, 1973; Fleis, 1969; Wampold & Serlin, 2000). Olejnik and Algina (2003) proposed the generalized eta-squared effect size, whose value is not influenced by the study design. The generalized eta-squared statistic is estimated using

$$\eta^2_G = \frac{SS_{effect}}{\delta(SS_{effect}) + \sum_{Meas} SS_{Meas} + \sum_K SS_K}$$

Where

$SS_{effect}$  is the sum of squares for the effect of interest,

$\delta = 1$  if the effect is a manipulated factor (and is zero otherwise),

$SS_{Meas}$  are the sums of squares for all sources of variance that involve measured factors (rather than manipulated factors) but do not include subjects, and

$SS_K$  are the sums of squares for all sources of variance that involve subjects.

Finally, the generalized omega-squared effect size is given by

$$\omega^2_G = \frac{SS_{effect} - (df_{effect})(MS_{error})}{\delta[SS_{effect} - (df_{effect})(MS_{error})] + \sum_{Meas} [SS_{Meas} - (df_{Meas})(MS_{Meas})] + (N)(MS_{s/Cells})}$$

where

$MS_{error}$  is the error mean square for testing the effect and  $df_{effect}$  are the degrees of freedom for the effect,  $MS_{Meas}$  and  $df_{Meas}$  are the analogous terms for each measured factor, and  $N$  is the total number of scores in the analysis.

## SOFTWARE LIMITATIONS

SAS does not provide the generalized eta-squared or generalized omega-squared effect sizes in either PROC ANOVA or PROC GLM (with the exception of single factor models, for which the generalized eta-squared statistic is equivalent to eta-squared). Thus, in order for SAS users to follow the APA Task Force recommendations for reporting effect sizes, extra work is required. For example, using the output from PROC ANOVA or PROC GLM, researchers can calculate these values either by hand or by outputting the ModelANOVA and OverallANOVA ODS tables and calculating the effect sizes through a data step. Unfortunately, the computational formulas for these statistics can be confusing with many ANOVA designs. Thus, in an effort to reduce the burden of calculating this effect size to accompany analysis of variance hypothesis test results, this paper provides a SAS macro to calculate generalized omega-squared for a variety of research designs.

Same as with the generalized eta-squared, generalized omega squared ( $\omega_G^2$ ) is intended for designs in which there is at least one categorical independent variable. For its estimation, it is important to understand the layout of the study factors, which helps shape and guide not only the analysis of variance but also the estimation of the generalized omega-squared effect size. One important distinction needs to be kept in mind when computing  $\omega_G^2$ . Research designs will involve “found”, measured factors which are individual differences (e.g., demographic variables) and “assigned”, manipulated factors by the researcher. That is, “variability in the data arise due to two sources of variance: manipulated factors and individual differences” (Olejnick & Algina, 2003; p. 436), or due to within-subjects and between-subjects factors. Table 1 presents the ANOVA models addressed in this paper and for which  $\omega_G^2$  is computed.

		Within-Subjects Factors		
		0	1	2
Between-Subjects Factors	0	N-A	✓	✓
	1	✓	✓	✓
	2	✓	✓	✓
	> 2	✓	✓	✓

**Table 1. ANOVA Models for Which  $\omega_G^2$  Is Computed.**

## MACRO GEN\_OMEGA2 DETAILS

The Gen\_Omega2 SAS macro computes the generalized omega-squared effect size (Olejnick & Algina, 2003), which is comparable across a wide variety of ANOVA research designs, utilizing data from PROC GLM ODS tables. The arguments to the macro include the name of the SAS dataset and the model ANOVA elements to be specified according to the research design (i.e., class, model, repeated, measured), as well as the specification of the dependent variable Y.

```
%macro GEN_OMEGA2 (data = _last_, class = , model = NONE, repeated = NONE, measured = NONE,
dependent = Y);
* extracts measured variables;
%if &measured ne NONE %then %do;
  data Vnames;
    noblank = 0; numchars = 0; vseq = 0; varnames = symget('measured');
    do i = 1 to length(varnames);
      if noblank = 0 then do;
        if substr(varnames,i,1) = ' ' then noblank = 0;
        if substr(varnames,i,1) ^= ' ' then noblank = i;
      end;
      if noblank ^= 0 then do;
        if substr(varnames,i,1) = ' ' | i = length(varnames) then do;
          numchars = i - noblank;
          if i = length(varnames) then numchars = numchars + 1;
          varname = substr(varnames,noblank,numchars);
          vseq = vseq + 1; noblank = 0; output;
        end;
      end;
    end;
  end;
```

```

end;
data Vnames; set Vnames; keep vseq varname;
data _null_; set Vnames end=lastrec;
    if lastrec then do; call symput('N_Measured',_n_); end;
run;
data Measured_variables; Merge_Var = 1; run;
%do jk = 1 %to &N_Measured;
    data build; set Vnames; if _n_ = &jk; measured_var&jk = varname; run;
    data Measured_variables; merge Measured_variables build; run;
%end;
%end; * End extracting measured variables;

**extracts names and number of levels of the within-subjects factor(s);
%if &repeated ne NONE %then %do;
data Vnames;
    noblank = 0; numchars = 0; vseq = 0;
    varnames = symget('repeated');
    do i = 1 to length(varnames);
        if noblank = 0 then do;
            if substr(varnames,i,1) = ' ' then noblank = 0;
            if substr(varnames,i,1) ^= ' ' then noblank = i;
        end;
        if noblank ^= 0 then do;
            if substr(varnames,i,1) = ' ' | i = length(varnames) then do;
                numchars = i - noblank;
                if i = length(varnames) then numchars = numchars + 1;
                varname = substr(varnames,noblank,numchars);
                vseq = vseq + 1; noblank = 0; output;
            end;
        end;
    end;
end;
data Vnames; set Vnames;
    do i = 1 to length(varname);
        if substr(varname,i,1) = ',' then substr(varname,i,1) = '';
    end;
    record_num = _N_;
    keep vseq varname record_num;
data _null_; set Vnames end=lastrec;
    if lastrec then do; call symput('N Repeated',_n_); end; run;
data Repeated_variables; Merge_Var = 1; run;
%do jk = 1 %to &N_Repeated;
    data build;
        set Vnames;
        if _n_ = &jk;
        if ROUND(record_num/2) ^= record_num/2 then do; * Odd numbered records;
            Repeated_var&jk = varname;
        end;
        if ROUND(record_num/2) = record_num/2 then do; * Even numbered records;
            Levels_var&jk = varname;
        end;
    data Repeated_variables; merge Repeated_variables build; run;
%end;
%end; * End parsing within subject factor;

** computes omega2 for between-subject factors;
%if &repeated = NONE %then %do;
    proc glm data = &data;
        class &class;
        model &dependent = &model / ss3;
        ods output ModelANOVA = two OverallANOVA = three ;
        run;
    data threeA; set three; if Source = 'Error'; MS W = MS; keep MS W ;
    data threeB; set three; if Source = 'Corrected Total'; N = DF + 1; keep N ;
    data threeC; set threeA; set threeB;
    %if &measured ne NONE %then %do;
        data two; set two; Merge_Var = 1;
        data twoA; merge Measured_variables two; by Merge_Var; Measured_Var = 0;
        %do jk = 1 %to &N_Measured;
            if FIND(Source,"**") ^= 0 then
                if INDEX(Source,TRIM(measured_var&jk)) ^= 0 then Measured_Var = 1;
            else;
            else if COMPARE(source,measured_var&jk,'i') = 0 then Measured_Var = 1;
        end;
    end;

```

```

%end;
delta = 1 - Measured_Var;
data TwoA;
    if _N_ = 1 then set ThreeC;    retain MS_W N; set TwoA;
    sigma_hat2 = (SS-(DF*MS_W))/N; *Num of omega_var effect ;
proc sort data = twoA out=sortedTwoA; by HypothesisType Measured_Var;

proc means noprint data = twoA; by HypothesisType Measured_Var; var sigma_hat2;
    output out = TwoB1 Sum = sigma_hat2 meas;
data TwoB; set TwoB1; if Measured_Var = 1; drop Measured_Var;
data two_back; merge TwoA TwoB; by HypothesisType;
data final; if _N_ = 1 then set threeC; retain MS_W N; set two_back;
Generalized_omega2 = sigma_hat2/(delta*sigma_hat2 + sigma_hat2_meas + MS_W);
%end; * End of measured factors loop;
%if &measured = NONE %then %do; *executes if have no measured factors;
    data final;
        if _N_ = 1 then set threeC; retain MS_W N; set two;
        delta = 1;
        sigma_hat2 = (SS-(DF*MS_W))/N;*Num of omega_var effect ;
        Generalized_omega2 = sigma_hat2/(delta*sigma_hat2 + MS_W);
    %end; * End of no measured block;
%end; *end of between-subject factor(s) only;

%if &repeated ^= NONE %then %do; *executes if have a within-subjects factor;
    %LET Model_Type = %INDEX(&model,NONE);
    %if %EVAL(&Model_Type) ^= 0 %then %do; *WITHIN-subject factors only;
        data temp; set &data; subject = _N_; merge_var = 1;
        data long; merge temp repeated variables; by merge_var;
        %if &N_repeated = 1 or &N_repeated = 2 %then %do;
            CALL SYMPUT ('R_Var1', repeated_var1);
        %end;
        %if &N_repeated > 2 %then %do;
            L_var1 = Levels_var2 + 0; L_var2 = Levels_var4 + 0;
            CALL SYMPUT ('R_Var1', repeated_var1); CALL SYMPUT ('R_Var2', repeated_var3);
        %end;
        data long; set long; array varlist [*] &dependent; my_index = 1;
        %if &N_repeated = 1 or &N_repeated = 2 %then %do;
            do &R_var1 = 1 to DIM(varlist);
                y = varlist[my_index]; output;
                my_index = my_index + 1;
            end;
        %end;
        %if &N_repeated > 2 %then %do;
            do &R_Var1 = 1 to L_var1;
            do &R_Var2 = 1 to L_var2;
                y = varlist[my_index]; output;
                my_index = my_index + 1;
            end;
            end;
        %end;
        %if &N_repeated = 1 or &N_repeated = 2 %then %do;
            proc glm data = long;
                class subject &R_Var1;
                model y = subject|&R_Var1 / ss3;
                ods output ModelANOVA = two OverallANOVA = three;
            run;
        %end;
        %if &N_repeated > 2 %then %do;
            proc glm data = long;
                class subject &R_Var1 &R_Var2;
                model y = subject|&R_Var1|&R_Var2 / ss3;
                ods output ModelANOVA = two OverallANOVA = three;
            run;
        %end;
        data threeB; set three; if Source = 'Corrected Total'; Big_N = DF + 1; keep Big_N ;
        data twoA; set two; if INDEX(Source,"subject") ^= 0;
        proc means noprint data = twoA; var SS DF;
            output out = Subjects SUM = SS_Subjects DF_Subjects;
        proc sort data=twoA; by descending Source;
        data twoS; set twoA; line = N ; MS Error = MS; keep line MS_Error;
        data twoB; set two; if INDEX(Source,"subject") = 0;
        proc sort data=twoB; by descending Source;

```

```

data twoN; set twoB; line = _N_; drop dependent;
data twoM; merge twoN twoS; by line; if MS = . then delete;
data final;
  if _N_ = 1 then set Subjects; retain SS_Subjects DF_Subjects;
  if _N_ = 1 then set threeB; retain Big_N;
  set twoM; Dependent = 'WithinSubject'; MS_Subj = SS_Subjects/DF_Subjects; delta = 1;
  Generalized_omega2 = ((SS - DF*MS_Error)/Big_N)/(delta*((SS - DF*MS_Error)/Big_N) +
MS_Subj);
%end; * End Model_type ^= 0 loop;

%if %EVAL(&Model_Type) = 0 %then %do; * Both between and within factors;
  proc glm data = &data;
    class &class;
    model &dependent = &model / ss3;
    repeated &repeated;
    ods output ModelANOVA = two OverallANOVA = three;
    run;

    * Obtain correct MS_Error for the effects;
    data get_MSE1; set two;
    if Dependent = 'WithinSubject' or Dependent = 'BetweenSubjects'; Line = _N_;
    proc sort data = get_MSE1; by descending Line;
    data get MSE; set get MSE1;
      if INDEX(Source,"Error") ^= 0 then MS_Error = MS; retain MS_Error;
    data twoA; set two; if INDEX(Source,"Error") ^= 0;
    proc means noprint data = twoA; var SS DF;
      output out = Subjects SUM = SS Subjects DF Subjects;
    data threeA; set three; if INDEX(Source,"Corrected Total") ^= 0;
    proc means noprint data = threeA; var DF;
      output out = Total_N sum = sum_DF N = howmany;

    %if &measured ^= NONE %then %do;
      data get_MSE; set get_MSE; Merge_Var = 1;
      data twoB; merge Measured variables get MSE; by Merge Var;
      if INDEX(Source,"Error") = 0 and (Dependent = 'WithinSubject' or
Dependent = 'BetweenSubjects');
        Measured_Var = 0;
      %do jk = 1 %to &N_Measured;
        if INDEX(Source,TRIM(measured_var&jk)) ^= 0 then Measured_Var = 1;
      %end;
      delta = 1 - Measured_Var; Sigma_Measured = SS - DF*MS_Error;
      proc sort data = twoB; by Measured_Var;
      proc means noprint data = twoB; by Measured_Var; var Sigma_Measured;
        output out = TwoC Sum = Sum_Measured;
      data TwoC; set TwoC; if Measured_Var = 1; drop Measured_Var;
      data two_back; if _N_ = 1 then set TwoC; retain Sum_Measured; set twoB;
      data final; if _N_ = 1 then set Subjects;
        if _N_ = 1 then set Total_N;
        retain SS_Subjects DF_Subjects sum_DF howmany;
        set two_back; Big_N = sum_DF + howmany;
        MS_subj = SS_Subjects/DF_Subjects;
        Sum_Measured = Sum_Measured/Big_N;
        Generalized_omega2 = ((SS - DF*MS_Error)/Big_N)/(delta*((SS -
DF*MS_Error)/Big_N) + Sum_Measured + MS_Subj);
      %end; * End of measured factors loop;

    %if &measured = NONE %then %do;
      data final; if _N_ = 1 then set Subjects;
      if _N_ = 1 then set Total_N; retain SS_Subjects DF_Subjects sum_DF howmany;
      set get_MSE;
      if INDEX(Source,"Error") = 0 and (Dependent = 'WithinSubject' or Dependent =
'BetweenSubjects');
        Big_N = sum_DF + howmany; MS_Subj = SS_Subjects/DF_Subjects; delta = 1;
        Generalized_omega2 = ((SS - DF*MS_Error)/Big_N)/(delta*((SS -
DF*MS_Error)/Big_N) + MS_Subj);
      %end; * End of no measured factors loop;
    %end; *End of between and within loop;
  %end; * End of within-subjects factors loop;

** Generate final report ;
data output; set final; getord = 0;
  do i = 1 to length(Source);
    if substr(Source,i,1) ^= '*' then getord = getord +1;

```

```

        else getord = getord +2;
    end;
    if dependent ne 'BetweenSubjects' and dependent ne 'WithinSubject' then
        dependent2 = 'BetweenSubjects';
    if dependent = 'BetweenSubjects' or dependent = 'WithinSubject' then
        dependent2 = dependent;
proc sort data = output; by dependent2 HypothesisType getord;
data ord;
    set output; by dependent2;
    file print notitles header =hd;
    if first.dependent2 then do; put @1 / dependent2 /; end;
    put @5 Source @30 SS 8.4 @45 HypothesisType @65 Generalized_omega2 8.4;
Return;
hd: put @1 'Generalized Omega-Squared Values' // @5 'Source' @30 'SS'
    @45 'Hypothesis Type' @65 'Generalized Omega2';
run;
%MEND GEN_OMEGA2;

```

## MACRO GEN\_OMEGA2 EXECUTION

In Example 1, the SAS data step creates a SAS data set called ONE, containing eight observations. Each observation has teachers' student ratings on three scales (cooperation, avoid, and peer) taken before and after an intervention (i.e., pretest-posttest). The macro Gen\_Omega2 is called twice to illustrate the computation of generalized omega squared,  $\omega_G^2$  for one within-subjects and for two-within subjects factors, respectively.

### EXAMPLE 1:

```

DATA ONE;
    input student coop1 coop2 avoid1 avoid2 peer1 peer2;
cards;
1 31 33 21 27 28 30
2 31 25 15 25 30 32
3 16 35 26 33 17 32
4 27 26 22 31 21 31
5 32 26 29 32 29 35
6 32 33 16 33 20 27
7 26 32 14 27 19 19
8 19 26 14 29 10 29
;
run;

**run the macro for one WITHIN-subject factor;

%gen_omega2 (data = one, class = , repeated = %str(domain 3),model = NONE, dependent =
coop1 avoid1 peer1 );

**run the macro for two WITHIN-subject factors;

%gen_omega2 (data = one, class = , repeated = %str(time 2, domain 3),model = NONE,
dependent = coop1 avoid1 peer1 coop2 avoid2 peer2);

```

Output 1a shows the  $\omega_G^2$  computed for one within-subjects factor main effect. Sums-of-squares for the within-subjects factor "domain" is also provided, along with the Hypothesis Type 3 value.

Generalized Omega-Squared Values			
Source	SS	Hypothesis Type	Generalized Omega2
WithinSubject			
domain	214.0833	3	0.1483

Output 1a. Example Output from Gen\_Omega2 Macro for One Within-Subjects Design

Output 1b shows the output provided by the macro, which was called to compute  $\omega_G^2$  for two within-subjects factors.

As shown in the output, the macro Gen\_Omega2 computed the generalized omega squared effect size for both main effects (time and domain) and for the interaction. It also provides the factors and interaction Sum-of-Squares (SS) and the Hypothesis Type 3 values.

Generalized Omega-Squared Values			
Source	SS	Hypothesis Type	Generalized Omega2
WithinSubject			
time	553.5208	3	0.2777
domain	105.0417	3	0.0393
time*domain	109.2917	3	0.0548

#### Output 1b. Example Output from Gen\_Omega2 Macro for Two Within-Subjects Design

In Example 2, the SAS data step creates the SAS data set ONE from 10 observations. Each observation records data corresponding to a score and the participant's group allocation, where "score" is the dependent variable and "group" is the between-subjects manipulated factor. The macro Gen\_Omega2 is invoked to illustrate the computation of the generalized omega squared,  $\omega_G^2$  for one between-subjects design.

#### EXAMPLE 2:

```
DATA ONE;
  input score group;
cards;
17 1
23 1
20 1
14 1
12 1
18 2
22 2
11 2
21 2
19 2
;
RUN;

** run the macro for one between-subjects factor which is not a measured variable;

%gen_omega2 (data = one, class = group, model = group, repeated = NONE, measured =
NONE, dependent = score);
```

Output 2 shows the computed of  $\omega_G^2$  for the group, between-subjects factor main effect, as well as the Sum-of-Squares and Hypothesis Type 3 values.

Generalized Omega-Squared Values			
Source	SS	Hypothesis Type	Generalized Omega2
BetweenSubjects			
group	2.5000	3	-0.0953

#### Output 2. Example Output from Gen\_Omega2 Macro for One Between-Subjects Design

In Example 3, the SAS data step creates SAS data set ONE from 12 observations, and it is used to show the computation of the generalized omega-squared effect size for three between-subjects and one within-subjects design, where "sex", is a between-subjects measured factor and "group" and "intensity" are between-subjects manipulated factors. X1 – X3 is the repeated measures within-subjects factor. The macro gen\_omega2 was invoked



to compute the  $\omega_G^2$  for this mixed design.

### EXAMPLE 3:

```
DATA ONE;
  input x1 x2 x3 group sex intensity;
  SUBJECT = _N_;
cards;
1 2 3 1 1 1
1 1 1 1 1 2
2 2 5 1 2 2
3 3 3 1 2 1
3 2 1 1 2 1
6 5 4 2 1 2
4 5 4 2 2 2
3 6 7 2 1 1
3 4 8 2 2 1
2 6 8 2 2 2
2 5 3 1 1 2
5 4 8 2 2 1
;
run;

** run the macro between-subjects and within-subjects factors;

%gen_omega2 (data = one, class = group sex intensity, model = group|sex|intensity,
repeated = %str(time 3), measured = sex, dependent = X1 - X3);
```

Output 3 shows the omega squared effect sizes are provided for all main effects and for all interaction effects. The between-subject and within-subject parts of the model are separated in the same way that PROC GLM separates these effects.

Generalized Omega-Squared Values			
Source	SS	Hypothesis Type	Generalized Omega2
BetweenSubjects			
sex	0.6806	3	-0.0307
group	58.6806	3	0.4290
intensity	0.0139	3	-0.0413
group*sex	1.1250	3	-0.0247
sex*intensity	0.0139	3	-0.0397
group*intensity	1.1250	3	-0.0253
group*sex*intensity	0.1250	3	-0.0382
WithinSubject			
time	15.1944	3	0.1370
time*sex	4.1944	3	0.0105
time*group	1.6944	3	-0.0238
time*intensity	2.0278	3	-0.0191
time*group*sex	2.5833	3	-0.0112
time*sex*intensity	9.3611	3	0.0801
time*group*intensity	8.0833	3	0.0592
time*group*sex*intensity	5.0833	3	0.0225

**Output 3. Example Output from Gen\_Omega2 Macro for Three Between-Subjects Factors and One Within-Subjects Factor**

## CONCLUSION

Measures of effect size quantify the magnitude of the difference between population means or the relationship between variables. Their value resides in the fact that they are independent of sample size and either as standardized mean differences or as the proportion of variance explained, effect size measures enhance the interpretability of results of applied research. Within analysis of variance models, eta squared and omega squared are commonly used measures of effect size; however, these effect size measures can be misinterpreted, or more importantly, their value is limited due to differences in research designs across studies. In those cases, both eta-squared and omega-squared effect sizes do not make up for the differences in the sources of variation, such as between-subject and within-subjects variability. Generalized eta-square and omega-squared account for variability of design factors; however, they are not provided directly in either PROC ANOVA or PROC GLM. In addition, their hand computation can be confusing. Analogous to the purpose of our previous paper presented in SGF 2013 (Rodríguez et al., 2013), the purpose of this paper was to provide SAS users with an easy to use SAS macro for computing the generalized omega-squared effect size for ANOVA designs, under several design layouts.

## REFERENCES

- Cohen, J. (1973). Eta-squared and partial eta-squared in fixed factor ANOVA designs. *Educational and Psychological Measurement*, 33, 107-112.
- Fleiss, J. L. (1969). Estimating the magnitude of experimental effects. *Psychological Bulletin*, 81, 273-276.
- Kromrey, J. D., & Bell, B. A. (2012). Effect size indices for dichotomized outcomes under variance heterogeneity: An empirical investigation of accuracy and precision. *JSM Proceedings. Social Statistical Section*, Alexandria, VA: American Statistical Association
- Olejnick, S., & Algina, J. (2003). Generalized eta and omega squared statistics: Measures of effect size for some common research designs. *Psychological Methods*, 8(4), 434-447.
- Rodriguez de Gil, P., Pham, T., Rasmussen, P., Kellermann, A., Romano, J., Chen, Y-H., Kromrey, J.D. (2013). GEN\_ETA2: A SAS® Macro for Computing the Generalized Eta-Squared Effect Size Associated with Analysis of Variance Models. *Proceedings of the Annual SAS Global Forum Conference*, Cary, NC: SAS Institute Inc.
- Thompson, B. (2007). Effect sizes, confidence intervals, and confidence intervals for effect sizes. *Psychology in the Schools*, 44(5), 423-432.
- Wampold, B. E., & Serlin, R. C. (2000). The consequences of ignoring a nested factor on measures of effect size in analysis of variance. *Psychological Methods*, 425-433.
- Wilkinson, L., & Task Force on Statistical Inference (1999). Statistical methods in psychological journals: Guidelines and Explanations. *American Psychologist*, 54, 594-604.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the first author at:

Anh P. Kellermann  
University of South Florida  
4202 East Fowler Ave., EDU 105  
Tampa, FL 33620  
Fax: (813) 974-4495  
Email: napham@mail.usf.edu

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.