

## Paper B2B5

**Help! I Need to Report a Crime! Why is PROC REPORT So Hard to Use?**

Dessa Overstreet, Decision Street, Atlanta, GA

**ABSTRACT**

Business analysts often need to create summarized reports for large amounts of data may be easily enhanced or changed based on business needs. PROC REPORT can replace cumbersome manual reporting and automatically generate sub-totals, percentages and summaries for reports as well as allow formatting for areas of interest within a report. This paper discusses how to prepare your data and to programmatically generate reports using PROC REPORT that contain information you would expect of any report including titles, footnotes, data and column formatting, percentages, averages and other statistical measures, sub-totals and grand totals. Let PROC REPORT be your partner in crime instead of criminally confusing!

**INTRODUCTION**

This paper focuses on using a dataset to create a preliminary results report using PROC REPORT. As a business analyst you would like to generate a report outlining credit results from an auto loan portfolio. Your manager wants you to provide segmented results along with counts, sub-totals for each lender decision and grand totals based for:

- Lender Decision - if the loan was approved, conditionally approved or declined
- New or Used – if the vehicle was new or used
- Loan or Lease – if the vehicle financing was for a loan or a lease
- Exclude any applications if no credit file is found and where it is unknown if the financing was for a loan or a lease and provide information on how many records are excluded

Your manager has a high level idea about what they want to see in the final report but will ask for various versions based on who the report is being provided to. Also, you want to verify your results make sense and want to use the report results to verify results.

**PREPARING YOUR DATA**

You read in the raw auto loan portfolio data and you now have a SAS dataset, WORK.BOGUS ready to begin writing the report. (1.input\_bogus\_report\_data.sas)

**SAMPLE RAW DATA**

This is what your raw data looks like (BOGUS\_Data\_SESUG.csv):

has\_ssn,lender\_decision,new\_used,loan\_lease,has\_credit\_file,credit\_score,trd\_high\_cred,total\_purchase\_amount,has\_auto\_trade,ever30,ever60,ever90plus

1,Approved,New,1,1,725,70000,75000,1,1,0,0

1,Approved,New,1,1,750,42500,50000,1,1,1,1

1,Approved,New,1,1,705,60000,55000,1,0,0,0

1,Approved,New,1,1,700,62500,60000,1,0,0,0

**CREATE FORMATS AND SIZE YOUR TOTAL POPULATION**

You notice that the loan\_lease variable is numeric instead of 'Loan' or 'Lease'. You create a format to use in your report so the information in this variable is easy to understand:

```
PROC FORMAT;
  VALUE ctype
    1='Loan'
    2='Lease'
    3='Unknown'
  ;
RUN;
```

Next, you create a global macro variable called `total_portfolio` to hold the total number of records in the data you were provided.

```
PROC SQL;
  SELECT NOBS INTO : total_portfolio
  FROM DICTIONARY.TABLES
  WHERE UPCASE(libname) = "WORK"
    and UPCASE(memname) = "BOGUS";
;
```

## CREATE REPORT DATA

You create your report dataset, `WORK.BOGUS_REPORT`, by applying the exclusions your manager has requested. Namely, if there is no credit report or financing is not identified as either a loan or a lease then you do not want to include this in the report.

```
DATA bogus_report;
  SET bogus (WHERE=((credit_score NE -1) AND (loan_lease NE 3)));
RUN;
```

Next, you create another global macro variable called `total_report` to hold the number of records that will be used in the report.

```
PROC SQL;
  SELECT NOBS INTO : total_report
  FROM DICTIONARY.TABLES
  WHERE UPCASE(libname) = "WORK"
    and UPCASE(memname) = "BOGUS_REPORT";
;
```

## CREATING THE REPORT

Now that you have the report dataset ready, you begin the ODS setup required to write your report using `PROC REPORT` and create the report as an HTML file.

### PREPARE OUTPUT DESTINATION (ODS)

You clear out the ODS HTML destination, set up the output to be landscape, and define the location and name of your report file.

```
ODS HTML CLOSE; ODS HTML;
OPTIONS ORIENTATION=landscape;
ODS HTML FILE = "C:\Users\dezza\Documents\SAS\SESUG\report_SESUG.HTML";
```

### Define Escape Character

You want to use a dagger (†) in your output and because you are using ODS, you must assign an escape character in order to use the dagger inline formatting function. In this case, you assign the pound symbol (#) as the escape character. Note: You can also use `UNICODE` function to generate the dagger sign but this is here for illustrative purposes in the event you need to define an escape character for your report.

```
ODS ESCAPECHAR '#';
```

## CREATE HEADERS AND FOOTERS FOR REPORT

You want to include some headers and footers in your report for clarity and readability:

- Current Date Header – right justified, smaller font
- General Header – center justified, larger font
- Two Footers with notes that are relevant to the report

```

** Title 1 generates and formats current system date. It is (J)ustified to
the right and is smaller font size (10pt);
TITLE1 J=r HEIGHT=10pt "%SYSFUNC (LEFT (%QSYSFUNC (DATE ( ) ,WORDDATE18.)) )";
TITLE2 J=c HEIGHT=14pt "COMPANY CONFIDENTIAL - AUTO SALES PORTFOLIO FOR XYZ
COMPANY";

FOOTNOTE2 HEIGHT=8pt "Remember FCRA Compliance Regulations";
FOOTNOTE3 HEIGHT=8pt "The following states are included in SESUG: FL, GA, AL,
MS, TN, SC, NC, VA, WV, KY";

```

## IDENTIFY THE COLUMNS YOU WANT IN YOUR REPORT

Now that you have the framework for your report, you begin the PROC REPORT procedure. The first step in generating a report after entering the procedure is to identify the columns (variables) you want in the report and in the order they should appear in the report. For this, you use the COLUMN command. You can create column labels for your report columns and you can also group columns together and apply a group heading to them.

For this report, you want to group your columns into five groups with group headings. You do this by using parentheses around the section of columns you want to group and include the group title in apostrophes at the beginning. You also want to create divider lines between sections of your report. These lines are called “dummy” in the code below.

It is useful, but not necessary, to organize your code so that it is easy to see how you want your output to look.

```

** NOWD prevents opening in interactive mode;
PROC REPORT DATA=bogus_report NOWD;

    COLUMN ('GROUP By Variables' lender_decision new_used loan_lease)
        dummy N=cnt cnt_pct dummy
    ('Vehicle Loan Metrics' credit_score trd_high_cred=trd_high_cred_SUM
        trd_high_cred=trd_high_cred_MEDIAN)
        dummy
    ('Vehicle Metrics' total_purchase_amount=ttl_purch_amt_SUM
        total_purchase_amount=ttl_purch_amt_med)
        dummy
    ('Financed Performance Metrics' has_auto_trade ever30 pct_ever30
        ever60 pct_ever60 ever90plus pct_ever90plus)
;

```

Output 1 shows what you want the report header output to look like:

| COMPANY CONFIDENTIAL - AUTO SALES PORTFOLIO FOR XYZ COMPANY |             |               |                      |                  |                     |                        |                       |                       |                        |                          |                              |                 |                   |                 |                    | August 25, 2013  |  |
|---|-------------|---------------|----------------------|------------------|---------------------|------------------------|-----------------------|-----------------------|------------------------|--------------------------|------------------------------|-----------------|-------------------|-----------------|--------------------|------------------|--|
| Group By Variables  |             |               | Vehicle Loan Metrics |                  |                     |                        |                       | Vehicle Metrics       |                        |                          | Financed Performance Metrics |                 |                   |                 |                    |                  |  |
| Lender Decision   | New or Used | Loan or Lease | Count                | Percent of Total | Median Credit Score | Total Dollars Financed | Median Finance Amount | Total Purchase Amount | Median Purchase Amount | Files With an Auto Trade | Total Ever 30 DPD            | Pct Ever 30 DPD | Total Ever 60 DPD | Pct Ever 60 DPD | Total Ever 90+ DPD | PCT Ever 90+ DPD |  |

### Output 1. Header Output from PROC REPORT

You can see the date and header title information and you can also see you have columns corresponding to all of the variables in your columns statement. You have four defined group headers: Group By Variables, Vehicle Loan Metrics, Vehicle Metrics and Financed Performance Metrics. The dummy columns are the grey vertical lines. We will define the individual variable/column labels in the next section.

- You may notice the column element (N=cnt) after the three group by variables. You can use the N statistics as a counter to count the number of observations in each subgroup. You can then use this count to create the “Percent of Total” column.
- If you want to use a variable to create more than one analysis statistic (sum, average, median), you reference the variable name from your dataset and then the name you want to use for the statistic. For example, you want to create the “Total Dollars Financed” and “Median Finance Amount” in the report. Both of these statistics use

the variable `credit_score_trd_high_cred`. Therefore, in the COLUMN definition statement, you have two references to the variable so PROC REPORT will know which statistic to put in which column of the report:

- o (1) `credit_score trd_high_cred=trd_high_cred_SUM` and (2) `trd_high_cred= trd_high_cred_MEDIAN`

## DEFINE THE COLUMNS YOU WANT IN YOUR REPORT

Now that you have identified the variables you want in your report, now you'll define how those variables will interact with one another and with the report. For this, you use the DEFINE command. The DEFINE command has the syntax:

```
DEFINE column_name / <define_type> <options_and_attributes>;
```

You should have one DEFINE statement for each variable identified in the COLUMN statement. Again, these should be in the order you want them to appear in the report. This report will use the following DEFINE types: GROUP, ANALYSIS, ORDER, and COMPUTED.

### GROUP Types

Your manager has asked that you create segmented reports for the lender decision, new/used vehicle and loan/lease financing. When you hear segmented this usually means that you will use the GROUP type. The first three columns in the report make up the "Group By Variables" group and are what your report will be segmented on.

```
** GROUP By Variables;
DEFINE lender_decision /GROUP 'Lender Decision';
DEFINE new_used /GROUP STYLE(COLUMN)={FOREGROUND=red} 'New or Used';
DEFINE loan_lease /GROUP ORDER=internal FORMAT=ctype. 'Loan or Lease';
```

Using the GROUP type will automatically generate sub segments of all the combinations of `lender_decision`, `new_used` and `loan_lease` so you don't have to create twelve separate reports (3decisions x 2car types x 2finance types = 12 combinations or segments).

- If you reference Output 1 you'll see the labels in the DEFINE statement appear in the report instead of the variable names. Labels are created by enclosing the text in apostrophes, 'Lender Decision'.
- You can change the way the column appears in the report by using the STYLE(COLUMN) option. The `new_used` column text color will be red in the report (FOREGROUND= red).
- You can use the ORDER option to change the order of the rows in the table. Here, you will use the internal order for `loan_lease`. This means you use the unformatted values of the variable to order them. Recall that `loan_lease` is numeric with 1=loan and 2=lease and since you want loan to appear first in the report before lease use the internal ordering. If you did not use this option lease would appear in the report before loan (alphabetical order).
- Because you want the words "Loan" or "Lease" to appear in the report, format the `loan_lease` variable using the ctype format created earlier.

### Creating Vertical Separator Columns

So that it is easy to proof your report, you want to create some colored columns in your report to separate variables groups. For this report it may not be necessary but for reports with large numbers of computed columns it may be useful to have a demarcation point between groups of columns.

```
** Dummy is gray vertical line;
DEFINE dummy /COMPUTED ' ' STYLE(COLUMN HEADER)={CELLWIDTH=2pt
BACKGROUND=gray99 BORDERCOLOR=gray99 FOREGROUND=gray99};
```

You create the dummy by using a COMPUTED define type. This means that you are creating a report item where there is no variable associated with it in the report dataset. Look and see for yourself. There is not variable named "dummy" in your `bogus_report` dataset. Note: A COMPUTED define type is different from a COMPUTE block.

You create the vertical column with no column label (' '), define the width to be smaller (CELLWIDTH=2pt) and make grey in color (fill, border and text).

## Calculating Percentage to Total

Percentages seem to be one of the most difficult aspects of PROC REPORT compared to other analytical tools. You have identified the N statistic as a way to count observations in groups and named it “cnt” in the COLUMN statement.

You now want to determine, for each subgroup or your GROUP columns, the percentage of that subgroup to the total. In order to do this, you must first COMPUTE the total number of records in your report dataset and then divide each subgroup total by the overall total to get the group percentage. You will use a COMPUTE block to do this.

COMPUTE blocks begin with COMPUTE and end with ENDCOMP;. The variable in the COLUMN and DEFINE statements are also referenced in the COMPUTE blocks to link the report statements together. The DEFINE statement has a type of COMPUTED.

```

** Count for Each Sub-GROUP - Lender_Decision + New_Used + Lease_Loan;
DEFINE cnt / FORMAT=COMMA12. STYLE (COLUMN)={FONT_WEIGHT=bold BACKGROUND=pink}
'Count';
** Calculate percent of total records (not by GROUP) - use COMPUTE BEFORE;
** Calculate total records;
    COMPUTE BEFORE;
        IF _BREAK_="_RBREAK_" THEN ttlcnt=cnt;
    ENDCOMP;
** Calculate percentage;
DEFINE cnt_pct /COMPUTED FORMAT=PERCENT6.1 STYLE (COLUMN)={FONT_WEIGHT=bold
FOREGROUND=red} 'Percent of Total';
    COMPUTE cnt_pct;
        cnt_pct=cnt/ttlcnt;
    ENDCOMP;

```

- The order is important for this code. You must first define the counter (cnt) before you create a total count (ttlcnt) and you must define cnt and ttlcnt before you can define the percentage, cnt\_pct.
- The 'Count' column is defined by using the N statistic, labeled as 'cnt' in the COLUMN statement. Results for this column are formatted with bold text font (FONT\_WEIGHT=bold) and pink column color (BACKGROUND=pink).
- In order to calculate the total records, ttlcnt, in the report dataset, you must utilize a COMPUTE BEFORE statement. This statement executes once at the beginning of the report and counts the total number of records in the dataset used to generate the report.
- The automatic temporary variable \_BREAK\_ is a variable created when you group variables in a report and is used to summarize group segments. The \_RBREAK\_ automatic temporary variable allows you to summarize across the entire report. Since we want the count of all records in the report, we use \_RBREAK\_.
- Using BEFORE or AFTER with COMPUTE means your computation is not a report item explicitly. Later, you can double check this against the global macro variable you created early, total\_report, and the grand total row created for the report. If you are computing a report item (column value) then you will use only COMPUTE. For example, the “Total Count” variable (ttlcnt) is used to calculate the “Percent of Total” variable (cnt\_pct). Ttlcnt is NOT a column in your report (so use COMPUTE BEFORE) whereas cnt\_pct IS a column in your report (so only use COMPUTE).
- Now you are ready to calculate the percentage of each subgroup to the total report records, cnt\_pct. Here you use a COMPUTE statement. You want the percentage to be formatted with one decimal place, the percent sign (%) and in bold, red font.

Output 2 shows a sample of what you want the report output to look like for these grouped and count columns:

| Group By Variables |             |               | Count | Percent of Total |
|--------------------|-------------|---------------|-------|------------------|
| Lender Decision    | New or Used | Loan or Lease |       |                  |
| Approved           | New         | Loan          | 9     | 8.1%             |
|                    |             | Lease         | 4     | 3.6%             |
|                    | Used        | Loan          | 13    | 12%              |
|                    |             | Lease         | 1     | 0.9%             |

**Output 2. Group By and Count Columns**

### Calculating Statistics

You want to calculate some statistics for your segmented and group report including the median and sum of several of your columns. You want to calculate the median credit score, the total dollars financed for vehicle loans, the median finance amount for vehicle loans, the total vehicle purchase amount (before adjustment of down payments, trade ins, etc.), and median vehicle purchase amount. For these types of calculations, you will use the ANALYSIS type.

```

** Vehicle Loan Metrics;
DEFINE credit_score /ANALYSIS MEDIAN FORMAT=3.0
  STYLE(HEADER)= {FLYOVER='Flyover!!'} 'Median Credit Score';
DEFINE trd_high_cred_SUM /ANALYSIS SUM FORMAT=DOLLAR20.0 'Total Dollars
  Financed';
DEFINE trd_high_cred_MEDIAN /ANALYSIS MEDIAN FORMAT=DOLLAR14.0 'Median
  Finance Amount';

** Vehicle Metrics;
DEFINE ttl_purch_amt_SUM /ANALYSIS SUM FORMAT=DOLLAR20.0 'Total Purchase
  Amount';
DEFINE ttl_purch_amt_MEDIAN /ANALYSIS MEDIAN FORMAT=DOLLAR19.0 'Median
  Purchase Amount';

```

- Since you only have one statistic you wanted to include for the credit\_score variable, no name association with the variable is necessary. You simply DEFINE the credit\_score column as ANALYSIS followed by the statistical measure you want (MEDIAN). You apply the appropriate label 'Median Credit Score' and format (FORMAT=3.0) since you want an integer value and credit scores are three digits.
- You can add text that appears when you hover over the column heading that will appear in the HTML output using the STYLE(HEADER)= option. This is useful if you want to include some help text for columns.
- For Finance and Purchase amounts, you want two statistics and you defined the column names in the COLUMN statement and associated them with the variables from your report dataset. Using the ANALYSIS type and desired statistics (SUM or MEDIAN in this case) you create the desired statistics.
- You can format dollars using the dollar sign (\$) and can round to whole dollars (FORMAT=DOLLAR19.0).

Output 3 shows a sample of what you want the report output to look like for these analysis and statistics columns:

| Group By Variables |             |               |       |                  | Vehicle Loan Metrics |                        |                       | Vehicle Metrics       |                        |
|--------------------|-------------|---------------|-------|------------------|----------------------|------------------------|-----------------------|-----------------------|------------------------|
| Lender Decision    | New or Used | Loan or Lease | Count | Percent of Total | Median Credit Score  | Total Dollars Financed | Median Finance Amount | Total Purchase Amount | Median Purchase Amount |
| Approved           | New         | Loan          | 9     | 8.1%             | 725                  | \$481,500              | \$60,000              | \$494,000             | \$55,000               |
|                    |             | Lease         | 4     | 3.6%             | 730                  | \$130,000              | \$33,500              | \$140,000             | \$33,500               |
|                    | Used        | Loan          | 13    | 12%              | 739                  | \$329,000              | \$26,500              | \$346,500             | \$25,000               |
|                    |             | Lease         | 1     | 0.9%             | 700                  | \$10,000               | \$10,000              | \$17,500              | \$17,500               |

**Output 3. Analysis and Statistics Columns**

### Calculating Percentage to Subsets of Report Data

If you want to calculate percentages based on a subpopulation of the total report population, you can use a combination of ANALYSIS and COMPUTE types.

First, you will calculate the total number of records you want to use as your denominator using the ANALYSIS SUM type. Do not forget to include this variable in the COLUMN statement!

Then, calculate the total number of records you want to use as the numerator using the ANALYSIS SUM type. Again, include this variable in your COLUMN statement!

Finally, you will calculate the percentage by dividing the numerator by the denominator and formatting the percentage accordingly. Of course, you included this variable in your COLUMN statement!

For your report, you want to calculate the percentage of late payments for car loans for your portfolio. However, not all vehicle loan applications result in a loan being issued. Up until now you could use the entire report dataset since the data was for vehicle financing applications. The denominator for these late percentages should be based on the total number of auto loans and not the total number of applications.

```

** Auto Trade Found on Credit FILE;
DEFINE has_auto_trade /ANALYSIS SUM /* NOPRINT */
  STYLE (COLUMN)={ FOREGROUND=red} 'Files with an Auto Trade';

** Credit Percentages;
DEFINE ever30 /ANALYSIS SUM /* NOPRINT */
  STYLE (COLUMN)={ FOREGROUND=red} 'Total Ever 30 DPD';

DEFINE pct_ever30 /COMPUTED FORMAT=PERCENT7.1 'Pct Ever 30 DPD';
  COMPUTE pct_ever30;
  pct_ever30=ever30.SUM/has_auto_trade.SUM;
  ENDCOMP;

```

- The first thing you do is create the denominator. In this case, you want the denominator of your percentage to be the total number of applications where an auto was financed on the credit file. You use ANALYSIS type to create the SUM of the has\_auto\_trade variable. This is a binary flag and is set to 1 if an auto trade was found on the credit file. Note: If you do not have these types of flags in your report dataset you may want to create them for ease of use in the report.
- Next, you create the numerator. Again, using ANALYSIS type to create the SUM of the ever30 binary variable (variable is 1 if payment has ever been 30 days late and 0 if payment has never been 30 days late).
- Finally, you create the percentage by using the COMPUTED type and a COMPUTE block to create the percentage of financed auto trades that have ever been 30 days past due, pct\_ever30. You use the suffix “.SUM” after each of the variables used as the numerator and denominator to tell the COMPUTE block that you want to use the calculated value of the variable and not the COLUMN value.

- The NOPRINT option gives you the ability to not print columns in the report. We will print all columns in the report so that we can confirm our calculations. However, in the final report you may only want the percentage calculation to appear and not the numerator and denominator values. You create the numerator and denominator columns with red font so you can confirm percentage calculations.
- You will repeat this process two more times to calculate the 'Pct Ever 60 DPD' and 'Pct Ever 90+ DPD' columns.

Output 4 shows a sample of what you want the report output to look like for these percentage columns:

| Financed Performance Metrics   |                         |                    |                         |                    |                          |                     |
|--------------------------------|-------------------------|--------------------|-------------------------|--------------------|--------------------------|---------------------|
| Files With<br>an Auto<br>Trade | Total<br>Ever 30<br>DPD | Pct Ever<br>30 DPD | Total<br>Ever 60<br>DPD | Pct Ever<br>60 DPD | Total<br>Ever 90+<br>DPD | PCT Ever<br>90+ DPD |
| 8                              | 3                       | 37.5%              | 1                       | 12.5%              | 1                        | 12.5%               |
| 4                              | 1                       | 25.0%              | 1                       | 25.0%              | 1                        | 25.0%               |
| 13                             | 2                       | 15.4%              | 1                       | 7.7%               | 0                        | 0.0%                |
| 1                              | 0                       | 0.0%               | 0                       | 0.0%               | 0                        | 0.0%                |

Output 4. Numerator, Denominator and Percentage Columns

### Creating Subtotals, Grand Totals and Horizontal Separator Rows

You're almost there! Your manager wants subtotals for each Lender Decision and grand totals for the overall report. Also, you want to insert a horizontal separator row (similar to the vertical 'Dummy') after each subtotal section for evaluation purposes.

```

**Create sub-totals AFTER each change in lender decision;
BREAK AFTER lender_decision / SUMMARIZE SUPPRESS
  STYLE(SUMMARY) = {FONT_WEIGHT=bold};
** Create horizontal dummy grey line after each change in Lender Decision
COMPUTE AFTER lender_decision /
  STYLE(LINES) = {FONT_SIZE=2pt BACKGROUND=gray99 FOREGROUND=gray99};
  LINE ' ';

  new_used = "Totals:";

ENDCOMP;

** RBREAK: Create grand total / summary across entire report;
RBREAK AFTER / SUMMARIZE
  STYLE(SUMMARY) = {FONT_WEIGHT=bold FONT_SIZE=3};
** Completes the PROC REPORT statement;
RUN;

```

- Use the BREAK AFTER statement with SUMMARIZE type to summarize the report at each change in lender decision. If you wanted a summary after each change in new\_used then you'd use that variable instead. You can use any of the variables of type GROUP for summaries depending on the granularity you need.
- Use a COMPUTE AFTER block to insert a label of "Totals:" on the summary lines and to create horizontal separator row between each lender decision.



## Creating Additional HTML Text

You can use ODS HTML TEXT= statements to place text in your HTML output. In this case, we want to put our macro variables at the very end of the process after all the report output (including the footers). Then we close out ODS and we're done!

```
** Place text where you place the code. In this case at end of report (even after footers);
```

```
ODS HTML TEXT= "#{dagger} Total Records in Bogus Portfolio :
```

```
%SYSFUNC(PUTN(&total_portfolio,comma18.))";
```

```
ODS HTML TEXT= "Total Records FROM Bogus Portfolio Used In Report :
```

```
%SYSFUNC(PUTN(&total_report,comma18.))";
```

```
** Clear ODS, TITLES and FOOTNOTES;
```

```
ODS HTML CLOSE;
```

```
ODS HTML; TITLE; FOOTNOTE;
```

## THE FINAL REPORT AND CODE

Output 5 shows what the entire report will look like:

Output 6 shows what the entire report will look like.

| COMPANY CONFIDENTIAL - AUTO SALES PORTFOLIO FOR XYZ COMPANY   |             |               |       |                  |                      |                        |                       |                       |                        |                              |                   |                 |                   |                 |                    |                  | August 25, 2013 |
|---|-------------|---------------|-------|------------------|----------------------|------------------------|-----------------------|-----------------------|------------------------|------------------------------|-------------------|-----------------|-------------------|-----------------|--------------------|------------------|-----------------|
| Group By Variables  |             |               |       |                  | Vehicle Loan Metrics |                        |                       | Vehicle Metrics       |                        | Financed Performance Metrics |                   |                 |                   |                 |                    |                  |                 |
| Lender Decision   | New or Used | Loan or Lease | Count | Percent of Total | Median Credit Score  | Total Dollars Financed | Median Finance Amount | Total Purchase Amount | Median Purchase Amount | Files With an Auto Trade     | Total Ever 30 DPD | Pct Ever 30 DPD | Total Ever 60 DPD | Pct Ever 60 DPD | Total Ever 90+ DPD | PCT Ever 90+ DPD |                 |
| Approved  | New         | Loan          | 9     | 8.1%             | 725                  | \$481,500              | \$60,000              | \$494,000             | \$55,000               | 8                            | 3                 | 37.5%           | 1                 | 12.5%           | 1                  | 12.5%            |                 |
|   |             | Lease         | 4     | 3.6%             | 730                  | \$130,000              | \$33,500              | \$140,000             | \$33,500               | 4                            | 1                 | 25.0%           | 1                 | 25.0%           | 1                  | 25.0%            |                 |
|   | Used        | Loan          | 13    | 12%              | 739                  | \$329,000              | \$26,500              | \$346,500             | \$25,000               | 13                           | 2                 | 15.4%           | 1                 | 7.7%            | 0                  | 0.0%             |                 |
|   |             | Lease         | 1     | 0.9%             | 700                  | \$10,000               | \$10,000              | \$17,500              | \$17,500               | 1                            | 0                 | 0.0%            | 0                 | 0.0%            | 0                  | 0.0%             |                 |
|   | Totals:     |               |       | 27               | 24%                  | 725                    | \$950,500             | \$32,500              | \$998,000              | \$32,000                     | 26                | 6               | 23.1%             | 3               | 11.5%              | 2                | 7.7%            |
| Cond Appr   | New         | Loan          | 13    | 12%              | 660                  | \$599,000              | \$42,500              | \$629,000             | \$50,000               | 11                           | 5                 | 45.5%           | 3                 | 27.3%           | 2                  | 18.2%            |                 |
|   |             | Lease         | 4     | 3.6%             | 680                  | \$130,000              | \$33,500              | \$140,000             | \$33,500               | 3                            | 1                 | 33.3%           | 1                 | 33.3%           | 0                  | 0.0%             |                 |
|   | Used        | Loan          | 20    | 18%              | 650                  | \$406,500              | \$21,250              | \$451,500             | \$21,000               | 20                           | 3                 | 15.0%           | 3                 | 15.0%           | 1                  | 5.0%             |                 |
|   |             | Lease         | 1     | 0.9%             | 625                  | \$24,000               | \$24,000              | \$19,000              | \$19,000               | 1                            | 0                 | 0.0%            | 0                 | 0.0%            | 0                  | 0.0%             |                 |
|   | Totals:     |               |       | 38               | 34%                  | 650                    | \$1,159,500           | \$27,500              | \$1,239,500            | \$30,000                     | 35                | 9               | 25.7%             | 7               | 20.0%              | 3                | 8.6%            |
| Decline   | New         | Loan          | 14    | 13%              | 510                  | \$765,000              | \$53,750              | \$790,000             | \$55,000               | 11                           | 11                | 100%            | 7                 | 63.6%           | 4                  | 36.4%            |                 |
|   |             | Lease         | 4     | 3.6%             | 555                  | \$145,000              | \$36,250              | \$155,000             | \$40,000               | 4                            | 2                 | 50.0%           | 2                 | 50.0%           | 1                  | 25.0%            |                 |
|   | Used        | Loan          | 24    | 22%              | 568                  | \$566,000              | \$26,000              | \$608,500             | \$25,000               | 18                           | 7                 | 38.9%           | 6                 | 33.3%           | 4                  | 22.2%            |                 |
|   |             | Lease         | 4     | 3.6%             | 543                  | \$95,000               | \$27,500              | \$102,500             | \$27,500               | 4                            | 1                 | 25.0%           | 0                 | 0.0%            | 0                  | 0.0%             |                 |
|   | Totals:     |               |       | 46               | 41%                  | 560                    | \$1,571,000           | \$28,750              | \$1,656,000            | \$32,500                     | 37                | 21              | 56.8%             | 15              | 40.5%              | 9                | 24.3%           |
|   |             |               | 111   | 100%             | 620                  | \$3,681,000            | \$27,500              | \$3,893,500           | \$30,000               | 98                           | 36                | 36.7%           | 25                | 25.5%           | 14                 | 14.3%            |                 |
| Remember FCRA Compliance Regulations<br>The following states are included in SESUG: FL, GA, AL, MS, TN, SC, NC, VA, WV, KY<br>† Total Records in Bogus Portfolio : 158<br>Total Records FROM Bogus Portfolio Used In Report : 111 |             |               |       |                  |                      |                        |                       |                       |                        |                              |                   |                 |                   |                 |                    |                  |                 |

Output 5. The Entire Report

## THE CODE

Included here is the full code for reference:

```
*****.
***** FILE HEADER and Global Macro Vars *****.
*****.
OPTIONS LS=max PS=max NOCENTER ERRORS=3 COMPRESS=yes NONUMBER NODATE;
*****.
***** Create FORMATS to Use in Report *****.
*****.
** This FORMAT labels the finance type as loan, lease or other for display;
PROC FORMAT;
  VALUE ctype
    1='Loan'
    2='Lease'
    3='Unknown'
  ;
RUN;
*****.
***** Generate Counts For DATASETS *****.
*****.
** Create a global macro variable called total_portfolio that counts the total records in the portfolio;
PROC SQL;
  SELECT NOBS INTO : total_portfolio
  FROM DICTIONARY.TABLES
  WHERE UPCASE(libname) = "WORK"
    and UPCASE(memname) = "BOGUS";
;
** Writes the macro variable value INTO the .log FILE;
%PUT Total Recs in Portfolio =&total_portfolio;
** Create the DATASET that you want to use for the report. You may have exclusions or other DATA
  that you don't want included in the report;
** For this report, credit_score=-1 means no credit report so exclude;
** Also, we want to look at loans and leases only and not "other" so exclude loan_lease = 3;
DATA bogus_report;
  SET bogus (WHERE=((credit_score NE -1) AND (loan_lease NE 3)));
RUN;

PROC SQL;
  SELECT NOBS INTO : total_report
  FROM DICTIONARY.TABLES
  WHERE UPCASE(libname) = "WORK"
    and UPCASE(memname) = "BOGUS_REPORT";
;
%PUT Total Recs with CNX Key =&total_report;
*****.
***** Generate Report using PROC REPORT *****.
*****.
ODS HTML CLOSE; ODS HTML;
OPTIONS ORIENTATION=landscape;
ODS HTML FILE = "C:\Users\dessal\Documents\SAS\SESUG\report_SESUG.HTML";
ODS ESCAPECHAR '#';
** Title 1 generates and formats current system date. It is (J)ustified to the right and is smaller font size (10pt);
TITLE1 J=r HEIGHT=10pt "%SYSFUNC(LEFT(%QSYSFUNC(DATE(),WORDDATE18.)))";
TITLE2 J=c HEIGHT=14pt "COMPANY CONFIDENTIAL - AUTO SALES PORTFOLIO FOR XYZ COMPANY";
FOOTNOTE2 HEIGHT=8pt "Remember FCRA Compliance Regulations";
FOOTNOTE3 HEIGHT=8pt "The following states are included in SESUG: FL, GA, AL, MS, TN, SC, NC, VA, WV, KY";
** NOWD prevents opening in interactive mode;
PROC REPORT DATA=bogus_report NOWD;
  COLUMN ('Group By Variables' lender_decision new_used loan_lease)
    dummy N=cnt cnt_pct dummy
    ('Vehicle Loan Metrics' credit_score trd_high_cred=trd_high_cred_SUM trd_high_cred=trd_high_cred_MEDIAN)
    dummy
    ('Vehicle Metrics' total_purchase_amount=ttl_purch_amt_SUM total_purchase_amount=ttl_purch_amt_MEDIAN)
    dummy
    ('Financed Performance Metrics' has_auto_trade ever30 pct_ever30 ever60 pct_ever60 ever90plus pct_ever90plus)
  ;
```

```

** GROUP By Variables;
DEFINE lender_decision /GROUP 'Lender Decision';
DEFINE new_used /GROUP 'New or Used';
DEFINE loan_lease /GROUP ORDER=internal FORMAT=ctype. 'Loan or Lease';
** Dummy is gray vertical LINE;
DEFINE dummy /COMPUTED '' STYLE(COLUMN HEADER)={CELLWIDTH=2pt
BACKGROUND=gray99 BORDERCOLOR=gray99 FOREGROUND=gray99};
** Count for Each Sub-GROUP - Lender_Decision + New_Used + Lease_Loan;
DEFINE cnt / FORMAT=COMMA12. STYLE(COLUMN)={FONT_WEIGHT=bold BACKGROUND=pink}
'Count';
** Calculate percent of total report records (not by GROUP) - use COMPUTE BEFORE;
COMPUTE BEFORE;
IF _BREAK_="_RBREAK_" THEN ttlcnt=cnt;
ENDCOMP;

DEFINE cnt_pct /COMPUTED FORMAT=PERCENT6.1 STYLE(COLUMN)={FONT_WEIGHT=bold
FOREGROUND=red} 'Percent of Total';
COMPUTE cnt_pct;
cnt_pct=cnt/ttlcnt;
ENDCOMP;
** Dummy is gray vertical LINE;
DEFINE dummy /COMPUTED '' STYLE(COLUMN)={CELLWIDTH=2pt BACKGROUND=gray99
BORDERCOLOR=gray99 FOREGROUND=gray99};
** Vehicle Loan Metrics;
DEFINE credit_score /ANALYSIS MEDIAN FORMAT=3.0 STYLE(HEADER)={FLYOVER='Flyover!!'}
'Median Credit Score';
DEFINE trd_high_cred_SUM /ANALYSIS SUM FORMAT=DOLLAR20.0 'Total Dollars
Financed';
DEFINE trd_high_cred_MEDIAN /ANALYSIS MEDIAN FORMAT=DOLLAR14.0 'Median
Finance Amount';
** Vehicle Metrics;
DEFINE dummy /COMPUTED '' STYLE(COLUMN)={CELLWIDTH=2pt BACKGROUND=gray99
BORDERCOLOR=gray99 FOREGROUND=gray99};
DEFINE ttl_purch_amt_SUM /ANALYSIS SUM FORMAT=DOLLAR20.0 'Total Purchase
Amount';
DEFINE ttl_purch_amt_MEDIAN /ANALYSIS MEDIAN FORMAT=DOLLAR19.0 'Median
Purchase Amount';
** Financed Performance Metrics;
DEFINE dummy /COMPUTED '' STYLE(COLUMN)={CELLWIDTH=2pt BACKGROUND=gray99
BORDERCOLOR=gray99 FOREGROUND=gray99};
** Auto Trade Found on Credit FILE;
DEFINE has_auto_trade /ANALYSIS SUM /* NOPRINT */ STYLE(COLUMN)={FOREGROUND=red}
'Files With an Auto Trade';
** Credit Percentages;
DEFINE ever30 /ANALYSIS SUM /* NOPRINT */ STYLE(COLUMN)={FOREGROUND=red} 'Total
Ever 30 DPD';
DEFINE pct_ever30 /COMPUTED FORMAT=PERCENT7.1 'Pct Ever 30 DPD';
COMPUTE pct_ever30;
pct_ever30=ever30.SUM/has_auto_trade.SUM;
ENDCOMP;

DEFINE ever60 /ANALYSIS SUM /* NOPRINT */ STYLE(COLUMN)={FOREGROUND=red} 'Total
Ever 60 DPD';
DEFINE pct_ever60 /COMPUTED FORMAT=PERCENT7.1 'Pct Ever 60 DPD';
COMPUTE pct_ever60;
pct_ever60=ever60.SUM/has_auto_trade.SUM;
ENDCOMP;

DEFINE ever90plus /ANALYSIS SUM /* NOPRINT */ STYLE(COLUMN)={FOREGROUND=red}
'Total Ever 90+ DPD';
DEFINE pct_ever90plus /COMPUTED FORMAT=PERCENT7.1 'PCT Ever 90+
DPD';
COMPUTE pct_ever90plus;
pct_ever90plus=ever90plus.SUM/has_auto_trade.SUM;
ENDCOMP;
** Create sub-totals AFTER each change in lender decision: Approved, Cond Appr, Decline;
BREAK AFTER lender_decision / SUMMARIZE SUPPRESS STYLE(SUMMARY) = {FONT_WEIGHT=bold
/*COLOR=magenta*/};
** Create horizontal dummy gray line after each change in Lender Decision;

```

```

COMPUTE AFTER lender_decision /
FOREGROUND=gray99;
LINE " ;
** Create label for sub-total rows;
new_used = "Totals:";
ENDCOMP;
** RBREAK : Create grand total / summary across entire report;
RBREAK AFTER / SUMMARIZE
/*COLOR=green*};
RUN;
** Place text where you place the code. In this case at end of report (even after footers);
ODS HTML TEXT= "{dagger} Total Records in Bogus Portfolio : %SYSFUNC(PUTN(&total_portfolio,comma18.))";
ODS HTML TEXT= "Total Records FROM Bogus Portfolio Used In Report : %SYSFUNC(PUTN(&total_report,comma18.))";
** Clear ODS, TITLES and FOOTNOTES;
ODS HTML CLOSE;
ODS HTML; TITLE; FOOTNOTE;

```

## CONCLUSION

This paper only begins to tap into the power of PROC REPORT and is provided to illustrate the solution to one business need. There are many topics that are not covered in this paper or are covered at a high level.

## REFERENCES

Carpenter, Art. 2007. *Carpenter's Complete Guide to the SAS REPORT Procedure*. Cary, NC, USA: SAS Institute Inc.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Dessa Overstreet  
 Enterprise: Decision Street, LLC.  
 Address: 5256 Peachtree Road  
 City, State ZIP: Atlanta, GA 30341  
 E-mail: [doverstreet@decisionstreet.com](mailto:doverstreet@decisionstreet.com) – or – [dessa.overstreet@gmail.com](mailto:dessa.overstreet@gmail.com)  
 Web: [www.decisionstreet.com](http://www.decisionstreet.com)

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.